



PRODUCT NAME

TM58XX (applicable to TM58P10, TM58P20, TM58PC10, TM58PC20, TM58PE10)

TITLE

Related to TO and PD of TM58XX

APPLICATION NOTE

- 1. Introduction of TO and PD bits**
- 2. DEMO Code**
- 3. Diagrams of the applicable circuits**

The BIT3 (**PD**) and BIT4 (**TO**) of the status register (STATUS: 03H) are not writable; instead, the bits that correspond to them are affected by changes in the status or execution of commands.

PD: It is a Power down flag bit: When power-on or by the CLRWDT instruction, it is “1”; when executing the SLEEP instruction, the bit is cleared to become “0”.

TO: It is a Time out flag bit. When power-on or executing the CLRWDT and SLEEP instruction, it is “1”; when the Occur WDT time-overflow, the bit is cleared to become “0”.

When power on or executing the CLRWDT command, both **TO** and **PD** are set as “1”, the changes in the status of these two bits are shown in the following table:

Event	TO	PD	Note
Power on reset	1	1	
WDT overflow	0	X	PD bit not affected
SLEEP instruction	1	0	
CLRWDT instruction	1	1	

The status of **TO** and **PD** after power on reset is shown in the table below:

TO	PD	Reset Event
0	0	WDT time out form sleep mode
0	1	WDT time out form normal mode
1	0	Input a “low” at Reset from sleep mode
1	1	Power on reset
X	X	Input a “low” at Reset from normal mode

The **TO** and **PD** bits are active low that can be used to determine different causes of reset. Events like the power on reset, Occur WDT time-overflow, or the RESET pin has been loaded with low voltage can all result in resetting.

Example: The RESET pin is connected to a button; the button is used as the ON/OFF button for the overall circuit.

Function: The pressing of the button will turn on the LED on the right side, when the button is pressed again, the LED will be turned off. If the button is not pressed again within one minute of turning on the LED, the program would turn off the LED automatically; if the button is pressed down now, the LED would be turned on again.

For details about the programming, please refer to "DEMO.ASM".

```

;ic:tm58p10
;config: fosc:nt(xt)
;        wdte:disable
;        cpt:on
;        type:general
;        lv:don't use
;=====
;register
;=====
iar      equ      00h
tmr0    equ      01h
pc      equ      02h
status  equ      03h
bsr     equ      04h
porta   equ      05h
portb   equ      06h
;=====
;status bit
;=====
c       equ      00h
dc      equ      01h
z       equ      02h
pd      equ      03h
to      equ      04h
pa0     equ      05h
pa1     equ      06h
pa2     equ      07h
;=====
count1  equ      07h
count2  equ      08h

        org      3ffh
        lgoto   main
        org      00h

```

```

main:    nop

        movla    00h
        iodir    porta
        movam    porta

        movla    02h
        select
        clrm    count1
        clrm    count2

        btmss    status,pd
        lgoto    start

s_sleep:
        bcm     porta,0    ; Turns off LED
        nop
        nop
        sleep           ;The RESET pin wakes up SLEEP mode and
                        ; begins to execute from RESET

        nop
        nop

start:
        clrwdt           ; Sets PD as 1, so that when the button is
                        ; pressed under the normal situation, the program
                        ; would enter SLEEP mode.

        bsm     porta,0    ; Turns on LED
        clrm    tmr0

wait:
        movm    tmr0,a
        xorla   0fah
        btmss   status,z
        lgoto   wait
        incm   count1,m
        movla   0fah
        subam   count1,a
        btmss   status,z
        lgoto   start
        clrm   count1
        incm   count2,m
        movla   78h
        subam   count2,a
        btmss   status,z
        lgoto   start
        clrm   count2
        lgoto   s_sleep
        end

```

Diagrams of the applicable circuits

